

**Ingeschreven cirkel:**

Het bepalen van de vergelijking van een ingeschreven cirkel bij een gegeven driehoek kan een wiskundige uitdaging zijn.

Er zijn aardig wat stappen te zetten voor je bij de oplossing bent.

**Stappen te zetten:**

- 1) **Bepaal de drie richtingsvectoren, normaalvectoren en vergelijkingen van de drie zijden.**
- 2) **Bepaal de bissectrices van twee hoeken** (zie opmerking 1)
- 3) **Bepalen snijpunt bissectrices, het punt  $M_c$**
- 4) **Bepaal de omtrek en oppervlakte van de driehoek** (zie opmerking 2 en 3)
- 5) **Bepaal de straal.**
- 6) **Geef formule:  $(x - xm)^2 + (y - ym)^2 = ST^2$**

**Opmerkingen:** Er is gewerkt met bepaalde formules. De juistheid van die formules wordt hier verder niet besproken omdat dat niet het doel van dit stuk is. Meer info daarover is te vinden in de bijgevoegde links of kan je navragen bij je wiskunde docent.

*Opmerking 1:* Afstand punten tot lijn:  
<https://nl.wikipedia.org/wiki/Afstand>

*Opmerking 2:* Oppervlakte en omtrek driehoek en straal ingeschreven cirkel:  
[https://nl.wikipedia.org/wiki/Ingeschreven\\_cirkel](https://nl.wikipedia.org/wiki/Ingeschreven_cirkel)

*Opmerking 3:* Je had ook het punt  $M_c$  en de formule van een bissectrice kunnen gebruiken. Daar is hier niet voor gekozen.


**Gegeven een driehoek  $A_1B_1C_1$**

Met :  $A_1$  basishoek linksonder  
 $B_1$  basishoek rechtsonder  
 $C_1$  tophoek

Bepaal de vergelijking van de ingeschreven cirkel.

In de tekst die hierna volgt zijn delen **rood** gedrukt.

Dat zijn de tussenstappen die zijn gebruikt in het programma voor de rekenmachine.



Uiteraard zou je het aantal tussenstappen kunnen verkleinen door verder te herleiden, maar door dat te doen zou het overzicht nog verder achteruit gaan dan nu al het geval is.

**Schrijf de drie zijden driehoek van de vorm:  $ax+by+c=0$**

Lijn  $\overrightarrow{A_1B_1}$ :  $Ax + By + C = 0$ :

Richtingsvector  $\overrightarrow{A_1B_1}$ :  $\begin{pmatrix} R \\ S \end{pmatrix} - \begin{pmatrix} P \\ Q \end{pmatrix} = \begin{pmatrix} R - P \\ S - Q \end{pmatrix}$

Normaalvector:  $\begin{pmatrix} Q - S \\ R - P \end{pmatrix}$  geeft:  $(Q - S)x + (R - P)y + C = 0$  ; door  $(P, Q)$

Drie variabelen worden dan:  **$A = (Q - S)$  ;  $B = (R - P)$  ;  $C = (-PA - QB)$**

Lijn  $\overrightarrow{A_1C_1}$ :  $Dx + Ey + F = 0$

Richtingsvector  $\overrightarrow{A_1C_1}$ :  $\begin{pmatrix} T \\ U \end{pmatrix} - \begin{pmatrix} P \\ Q \end{pmatrix} = \begin{pmatrix} T - P \\ U - Q \end{pmatrix}$

Normaalvector:  $\begin{pmatrix} Q - U \\ T - P \end{pmatrix}$  geeft:  $(Q - U)x + (T - P)y + F = 0$  ; door  $(P, Q)$

Drie variabelen worden dan:  **$D = (Q - U)$  ;  $E = (T - P)$  ;  $F = (-PD - QE)$**

Lijn  $\overrightarrow{B_1C_1}$ :  $Gx + Hy + I = 0$

Richtingsvector  $\overrightarrow{B_1C_1}$ :  $\begin{pmatrix} T \\ U \end{pmatrix} - \begin{pmatrix} R \\ S \end{pmatrix} = \begin{pmatrix} T - R \\ U - S \end{pmatrix}$

Normaalvector:  $\begin{pmatrix} S - U \\ T - R \end{pmatrix}$  geeft:  $(S - U)x + (T - R)y + I = 0$  ; door  $(T, U)$

Drie variabelen worden dan:  **$G = (S - U)$  ;  $H = (T - R)$  ;  $I = (-TG - UH)$**

**Bissectrice lijn  $\overrightarrow{A_1B_1}$  en  $\overrightarrow{A_1C_1}$ :**

$$\frac{|Ax+By+C|}{\sqrt{A^2+B^2}} = \frac{|Dx+Ey+F|}{\sqrt{D^2+E^2}} \text{ met } \mathbf{K} = \sqrt{A^2 + B^2} \text{ en } \mathbf{L} = \sqrt{D^2 + E^2}$$

$$L(Ax + By + C) = K(Dx + Ey + F) \quad \text{óf} \quad L(Ax + By + C) = -K(Dx + Ey + F)$$

$$x(AL - DK) + y(BL - EK) = FK - CL \quad \text{óf} \quad x(\mathbf{AL} + \mathbf{DK}) + y(\mathbf{BL} + \mathbf{EK}) = (-\mathbf{FK} - \mathbf{CL})$$

**Bissectrice lijn  $\overrightarrow{A_1B_1}$  en  $\overrightarrow{B_1C_1}$ :**

$$\frac{|Ax+By+C|}{\sqrt{A^2+B^2}} = \frac{|Gx+Hy+I|}{\sqrt{G^2+H^2}} \text{ met } \mathbf{K} = \sqrt{A^2 + B^2} \text{ en } \mathbf{M} = \sqrt{G^2 + H^2}$$

$$M(Ax + By + C) = K(Gx + Hy + I) \quad \text{óf} \quad M(Ax + By + C) = -K(Gx + Hy + I)$$

$$x(\mathbf{AM} - \mathbf{GK}) + y(\mathbf{BM} - \mathbf{HK}) = (\mathbf{IK} - \mathbf{CM}) \quad \text{óf} \quad x(\mathbf{AM} + \mathbf{GK}) + y(\mathbf{BM} + \mathbf{HK}) = (-\mathbf{IK} - \mathbf{CM})$$

De twee blauw gedrukte formules zijn relevant en worden hier verder gebruikt.

**Snijden van de twee bissectrices:**

$$x(AL + DK) + y(BL + EK) = (-FK - CL)$$

$$x(AM - GK) + y(BM - HK) = (IK - CM)$$

$$x_m: V = \frac{((BM-HK)(-FK-CL)-(BL+EK)(IK-CM))}{((AL+DK)(BM-HK)-(AM-GK)(BL+EK))}$$

$$y_m: W = \frac{((-FK-CL)-V(AL+DK))}{(BL+EK)}$$

Middelpunt ingeschreven cirkel:  $M_c(V, W)$

**Straal cirkel:**

$$|A_1B_1| = \sqrt{(R - P)^2 + (S - Q)^2} = \sqrt{B^2 + A^2} = K$$

$$|B_1C_1| = \sqrt{(R - T)^2 + (S - U)^2} = \sqrt{H^2 + G^2} = M$$

$$|A_1C_1| = \sqrt{(T - P)^2 + (U - Q)^2} = \sqrt{E^2 + D^2} = L$$

$$\text{Omtrek: } N = \frac{(K+L+M)}{2}$$

$$\text{Oppervlakte: } O = \sqrt{N(N - K)(N - L)(N - M)}$$

$$\text{Straal}^2: J = \left(\frac{O}{N}\right)^2$$

**Ingeschreven cirkel wordt dan:**

$$(x - V)^2 + (y - W)^2 = J$$

In het programma zijn bijna al de letters A t/m Z gebruikt om data of delen van de berekeningen op te slaan. Je kan diezelfde letters terugvinden in de rood-gedrukte stukken tekst.

**Program:CIRKELIN**

```
Disp "INGESCHREVEN CIRKEL"  
Disp "VAN DRIEHOEK ABC MET"  
Disp "A1 BASISHOEK LINKS"  
Disp "B1 BASISHOEK RECHTS"  
Disp "C1 TOPHOEK"  
Disp "A1(P,Q)"  
Prompt P,Q  
Disp "B1(R,S)"  
Prompt R,S  
Disp "C1(T,U)"  
Prompt T,U  
If P=R and P=T or Q=S and Q=U  
Then  
Goto Z  
Else  
PlotsOff  
ClrAllLists  
{P,R,T,P}→L1  
{Q,S,U,Q}→L2  
Q-S→A  
R-P→B  
-1PA-QB→C  
Q-U→D  
T-P→E  
-1PD-QE→F  
S-U→G  
T-R→H  
-1TG-UH→I  
 $\sqrt{A^2+B^2}$ →K  
 $\sqrt{D^2+E^2}$ →L  
 $\sqrt{G^2+H^2}$ →M  
((BM-KH)(-1KF-LC)-(BL+KE)(KI-MC))/((AL+DK)(BM-KH)-(AM-KG)(BL+EK))→V  
((-1KF-LC)-V(AL+DK))/(BL+EK)→W  
(K+L+M)/2→N  
 $\sqrt{N(N-K)(N-L)(N-M)}$ →O  
(O/N)2→J  
Disp "(X-XM)2+(Y-YM)2=ST2"  
Disp "MET XM=",V ▶ Frac  
Disp "YM=",W ▶ Frac  
Disp "ST2=",J ▶ Frac  
Pause  
ZSquare  
Plot1(xyLine,L1,L2)  
Circle(V,W, $\sqrt{J}$ )  
Stop
```

Als punten op 1 lijn, dan kan het niet

Zet eventuele andere plots uit.

Gebruik x- en y-waarden voor tekenen driehoek.

Zet window op square. Zo wordt de cirkel ook rond.

Hier plot je de driehoek ABC

Plot van de ingeschreven cirkel


**Lbl Z**  
**Disp "PUNTEN OP EEN LIJN"**  
**Disp "KAN NIET"**  
**Stop**

**Voorbeelden:**

Alleen de coördinaten van de hoekpunten gebruikt als invoer bij het programma!  
Zie de screenshots voor de resultaten


Geef formule van ingeschreven cirkel door de punten:

$A_1(-7, -3); B_1(4, -2); C_1(-6,5)$



Geef formule van ingeschreven cirkel door de punten:

$A_1(-2, -3); B_1(4, -3); C_1(-2,5)$



Geef formule van ingeschreven cirkel door de punten:

$A_1(-7, -3); B_1(1, -1); C_1(8,3)$

